


```

; Most of these should work on even non-IDE hard disks.
; This code is for reading, the code for writing is the next article.

mov     dx,1f6h           ;Drive and head port
mov     al,0a0h          ;Drive 0, head 0
out     dx,al

mov     dx,1f2h          ;Sector count port
mov     al,1             ;Read one sector
out     dx,al

mov     dx,1f3h          ;Sector number port
mov     al,1             ;Read sector one
out     dx,al

mov     dx,1f4h          ;Cylinder low port
mov     al,0             ;Cylinder 0
out     dx,al

mov     dx,1f5h          ;Cylinder high port
mov     al,0             ;The rest of the cylinder 0
out     dx,al

mov     dx,1f7h          ;Command port
mov     al,20h           ;Read with retry.
out     dx,al
still_going:
in      al,dx
test    al,8             ;This means the sector buffer requires
                        ;servicing.
jz      still_going      ;Don't continue until the sector buffer
                        ;is ready.

mov     cx,512/2         ;One sector /2
mov     di,offset buffer
mov     dx,1f0h          ;Data port - data comes in and out of here.
rep     insw

; -----
mov     ax,201h          ;Read using int13h then compare buffers.
mov     dx,80h
mov     cx,1
mov     bx,offset buffer2
int     13h

mov     cx,512
mov     si,offset buffer
mov     di,offset buffer2
repe   cmpsb
jne     failure

```

```

    mov     ah,9
    mov     dx,offset readmsg
    int     21h
    jmp     good_exit
failure:
    mov     ah,9
    mov     dx,offset failmsg
    int     21h
good_exit:
    mov     ax,4c00h        ;Exit the program
    int     21h

    readmsg db     'The buffers match.  Hard disk read using ports.$'
    failmsg db     'The buffers do not match.$'

buffer db     512 dup ('V')
buffer2 db    512 dup ('L')

;
;     Writing to the hard disk using the ports!     by qark
;     +-----+
;
; The only differences between reading and writing using the ports is
; that 30h is sent to the command register, and instead of INSW you
; OUTSW.
;

    mov     dx,1f6h        ;Drive and head port
    mov     al,0a0h        ;Drive 0, head 0
    out     dx,al

    mov     dx,1f2h        ;Sector count port
    mov     al,1           ;Write one sector
    out     dx,al

    mov     dx,1f3h        ;Sector number port
    mov     al,1           ;Wrote to sector one
    out     dx,al

    mov     dx,1f4h        ;Cylinder low port
    mov     al,0           ;Cylinder 0
    out     dx,al

    mov     dx,1f5h        ;Cylinder high port
    mov     al,0           ;The rest of the cylinder 0
    out     dx,al

    mov     dx,1f7h        ;Command port
    mov     al,30h        ;Write with retry.
    out     dx,al

```

```

oogle:
    in     al,dx
    test  al,8           ;Wait for sector buffer ready.
    jz    oogle

    mov   cx,512/2      ;One sector /2
    mov   si,offset buffer
    mov   dx,1f0h      ;Data port - data comes in and out of here.
    rep  outsw         ;Send it.

; -----

    mov   ax,201h      ;We'll read in sector 1 using
    mov   bx,offset buffer2 ;int13h and see if we are successful.
    mov   cx,1
    mov   dx,80h
    int   13h

    mov   cx,512
    mov   si,offset buffer
    mov   di,offset buffer2
    repe cmpsb        ;Compare the buffers.
    jne  failure

    mov   ah,9
    mov   dx,offset write_msg
    int   21h
    jmp  w_exit
failure:
    mov   ah,9
    mov   dx,offset fail
    int   21h

w_exit:
    mov   ax,4c00h     ;Exit the program
    int   21h

    write_msg    db    'Sector one written to using the ports, OH NO! there goes
XP.$'
    fail         db    'Writing using ports failed.$'

buffer  db    512 dup ('A')
buffer2 db    512 dup ('D')

```